Open-Economy Production Network New Keynesian Model

Model Equations, Calibration, and Dynare Implementation

July 10, 2026

Contents

1 Model Overview

This model extends **?** to a small open economy. The closed-economy model shows that, with multiple sectors and intermediate-input linkages, productivity shocks generate an inflation–output tradeoff for every standard price index except a unique *divine coincidence* (DC) index, which weights sectors by Domar weights scaled by price stickiness. We add three open-economy ingredients: (i) imported intermediate inputs (with home/foreign Armington nesting at each production node), (ii) imported final consumption goods, and (iii) a single internationally traded bond with a debt-elastic premium.

The key result is that the exchange rate enters sectoral marginal costs exactly like a productivity shock scaled by the import-content matrix $\mathbf{\Omega}^F$. By embedding import nodes as fully flexible *upstream sectors* ($\widehat{\delta}_M = 1$) in an augmented network, the DC index still exists and is immune to both domestic TFP shocks and FX/import-price shocks. A float targeting the DC index implements zero output gap and fully insulates the economy from external shocks via exchange-rate adjustment. A hard peg forces the full adjustment onto domestic inflation and output.

The model is calibrated to a **three-sector oil-exporting economy**: an upstream Resource sector (import-intensive inputs, most output exported), a Manufacturing sector (uses resources as intermediates), and a Services sector (sells mainly to domestic consumers). Three shocks are studied—an import price shock (ε^{pF}), a domestic TFP shock (ε^{a}), and a foreign demand shock (ε^{D})—under three policy regimes: free float, hard peg, and managed float.

2 Environment

2.1 Indexing and notation

There are $N = 3$ domestic sectors indexed $i, j \in \{1, 2, 3\}$ (Resource, Manufacturing, Services). Lower-case hats denote log-deviations from the non-stochastic steady state; tildes denote gaps (sticky-price minus flex-price). $E_t[\cdot]$ is the rational expectation conditional on information at t.

Network matrices.

$$
\begin{aligned}
\Omega^H_{ij} &= \text{domestic cost share of input } j \text{ in sector } i\text{'s costs},\\
\Omega^F_{ij} &= \text{imported cost share of input } j \text{ in sector } i\text{'s costs},\\
\alpha_i &= 1 - \sum_j (\Omega^H_{ij} + \Omega^F_{ij}) = \text{labour share in sector } i.
\end{aligned}
$$

The domestic Leontief inverse is $\mathbf{L}^H \equiv (\mathbf{I} - \mathbf{\Omega}^H)^{-1}$. The **domestic supplier centrality** (open-economy Domar weight) is

$$\lambda_{D,i} \equiv \left[\boldsymbol{\beta}^{H\top}\mathbf{L}^H\right]_i, \tag{1}$$

where $\boldsymbol{\beta}^H = (\beta^H_1, \beta^H_2, \beta^H_3)^\top$ are domestic final consumption shares.

Calvo adjustment. δ_i is the probability that a firm in sector i resets its price each period. Following **?**, define the adjusted parameter

$$\widehat{\delta}_i \equiv \frac{\delta_i\big(1-\beta(1-\delta_i)\big)}{1-\beta\delta_i(1-\delta_i)}. \tag{2}$$

2.2 Households

The representative household maximises

$$\mathbb{E}_0 \sum_{t=0}^{\infty} \beta^t \left[\frac{C_t^{1-\gamma}}{1-\gamma} - \frac{L_t^{1+\varphi}}{1+\varphi}\right]. \tag{3}$$

Aggregate consumption is a CES composite of a domestic bundle C^H_t and an imported bundle C^F_t:

$$C_t = \left[\omega^{1/\eta}(C^H_t)^{\frac{\eta-1}{\eta}} + (1-\omega)^{1/\eta}(C^F_t)^{\frac{\eta-1}{\eta}}\right]^{\frac{\eta}{\eta-1}}, \tag{4}$$

where ω is home bias in final consumption and η is the home–foreign elasticity. The domestic bundle aggregates across sectors with shares $\boldsymbol{\beta}^H$; the foreign bundle aggregates across M imported goods with shares $\boldsymbol{\mu}$ and elasticity σ.

The budget constraint is

$$P^C_t C_t + \frac{B_{t+1}}{1+i_t} + S_t \frac{B^*_{t+1}}{1+i^{*,\text{eff}}_t} \le W_t L_t + \Pi_t - T_t + B_t + S_t B^*_t, \tag{5}$$

where S_t is the nominal exchange rate (home currency per unit of foreign), B_t is a domestic bond, B^*_t is a foreign-currency bond, and Π_t are firm profits rebated lump sum.

The effective world rate includes a **debt-elastic premium**:

$$i^{*,\text{eff}}_t = i^* - \psi\left(\frac{S_t B^*_t}{P^C_t} - \bar{b}^*\right), \quad \psi > 0, \tag{6}$$

which ensures stationarity of net foreign assets (requires $\psi > 1/\beta - 1$).

2.3 Firms

Each domestic sector i has a unit mass of monopolistically competitive firms with CRS production:

$$Y_{ift} = A_{it} F_i\big(L_{ift}, \{X_{ijft}\}_{j=1}^N\big). \tag{7}$$

The intermediate input bundle X_{ijft} is an Armington composite of domestic and imported varieties of good j:

$$X_{ijft} = \left[v_{ij}^{1/\xi} (X_{ijft}^H)^{\frac{\xi-1}{\xi}} + (1 - v_{ij})^{1/\xi} (X_{ijft}^F)^{\frac{\xi-1}{\xi}} \right]^{\frac{\xi}{\xi-1}}. \tag{8}$$

Cost minimisation gives nominal marginal cost

$$MC_{it} = \frac{W_t^{\alpha_i} \prod_j \tilde{P}_{ijt}^{\omega_{ij}}}{A_{it}}, \tag{9}$$

where $\tilde{P}_{ijt}$ is the Armington price of the input bundle (weighted average of domestic price P_{jt} and foreign price $S_t P_{F,jt}^*$). Firms face Calvo pricing: fraction δ_i reset optimally each period.

2.4 Policy and shocks

The central bank sets the nominal interest rate i_t under one of three regimes (Section **??**). Exogenous shocks are AR(1) processes.

3 Log-Linearised System

All variables are log-deviations from the zero-inflation steady state. The full system for the three-sector model is listed below; these are the equations entered directly into Dynare.

3.1 Sectoral Phillips curves

For each sector $i \in \{1, 2, 3\}$, the Calvo pricing condition yields:

$$\boxed{\pi_{it} = \beta(1 - \widehat{\delta}_i)\, \mathbb{E}_t \pi_{i,t+1} + \widehat{\delta}_i\, \widetilde{mc}_{it}} \tag{10}$$

where the **real marginal cost gap** is

$$\widetilde{mc}_{it} = \alpha_i \widehat{w}_t + \sum_j \omega_{ij}^H\, p_{jt} + \omega_i^F (\widehat{s}_t + \widehat{p}_t^{F*}) - \widehat{a}_{it} - p_{i,t-1} \tag{11}$$

with $\omega_i^F \equiv \sum_j \Omega_{ij}^F$ (sector i's total import share in costs) and $p_{jt} = p_{j,t-1} + \pi_{jt}$.

Sector 1 — Resource: no domestic inputs ($\Omega_{1j}^H = 0$), 30% imported machinery.

$$\pi_{1t} = \beta(1 - \widehat{\delta}_1)\mathbb{E}_t \pi_{1,t+1} + \widehat{\delta}_1 \left[\alpha_1 \widehat{w}_t + \omega_1^F (\widehat{s}_t + \widehat{p}_t^{F*}) - \widehat{a}_{1t} - p_{1,t-1} \right] \tag{12}$$

Sector 2 — Manufacturing: 20% from domestic Resource, 10% imported.

$$\pi_{2t} = \beta(1 - \widehat{\delta}_2)\mathbb{E}_t \pi_{2,t+1} + \widehat{\delta}_2 \left[\alpha_2 \widehat{w}_t + \omega_{21}^H p_{1t} + \omega_2^F (\widehat{s}_t + \widehat{p}_t^{F*}) - \widehat{a}_{2t} - p_{2,t-1} \right] \tag{13}$$

Sector 3 — Services: 25% from domestic Manufacturing, 5% imported.

$$\pi_{3t} = \beta(1 - \widehat{\delta}_3)\mathbb{E}_t \pi_{3,t+1} + \widehat{\delta}_3 \left[\alpha_3 \widehat{w}_t + \omega_{32}^H p_{2t} + \omega_3^F (\widehat{s}_t + \widehat{p}_t^{F*}) - \widehat{a}_{3t} - p_{3,t-1} \right] \tag{14}$$

3.2 Price level laws of motion

$$p_{it} = p_{i,t-1} + \pi_{it}, \qquad i = 1, 2, 3. \tag{15}$$

3.3 Real wage (labour supply)

Log-linearising the intratemporal optimality condition and using Lemma 2 (output gap = employment gap):

$$\widehat{w}_t = \widehat{p}_t^C + (\gamma + \varphi)\,\widetilde{y}_t, \tag{16}$$

where the **consumer price level** in log-deviations is

$$\widehat{p}_t^C = \sum_i \beta_i^H\, p_{it} + \beta^{F,\text{tot}}(\widehat{s}_t + \widehat{p}_t^{F*}). \tag{17}$$

3.4 CPI inflation

$$\pi_t^C = \sum_i \beta_i^H\, \pi_{it} + \beta^{F,\text{tot}}\big[(\widehat{s}_t + \widehat{p}_t^{F*}) - (\widehat{s}_{t-1} + \widehat{p}_{t-1}^{F*})\big]. \tag{18}$$

3.5 Divine coincidence (DC) index

$$\pi_t^{DC} = \sum_i w_i\, \pi_{it}, \qquad w_i = \frac{\lambda_{D,i}(1-\widehat{\delta}_i)/\widehat{\delta}_i}{\sum_k \lambda_{D,k}(1-\widehat{\delta}_k)/\widehat{\delta}_k}. \tag{19}$$

By Proposition **??**, $\pi_t^{DC} = \rho \mathbb{E}_t \pi_{t+1}^{DC} + \kappa^{DC}\widetilde{y}_t$ with no residual term.

3.6 IS curve

Log-linearising the Euler equation and imposing market clearing:

$$\widetilde{y}_t = \mathbb{E}_t \widetilde{y}_{t+1} - \frac{1}{\gamma}\big(\widehat{\imath}_t - \mathbb{E}_t \pi_{t+1}^C - r_{t+1}^{nat}\big) + \nu_D \widehat{D}_t^*, \tag{20}$$

where r_{t+1}^{nat} is the natural (flex-price) real interest rate and $\nu_D > 0$ captures the direct effect of foreign demand on domestic output.

3.7 Uncovered interest parity (UIP)

$$\widehat{\imath}_t = \widehat{\imath}^* - \psi\, \widehat{b}_t^* + \mathbb{E}_t \widehat{s}_{t+1} - \widehat{s}_t. \tag{21}$$

3.8 Net foreign assets (NFA)

The balance-of-payments identity in log-linear form:

$$\widehat{b}_{t+1}^* = \underbrace{\left(\frac{1}{\beta} - \psi\right)}_{<1} \widehat{b}_t^* + \kappa_{EX}\left(-\theta^* \widehat{s}_t + \widehat{D}_t^*\right) - \kappa_{IM}\left(\widehat{p}_t^{F*} + \widetilde{y}_t\right), \tag{22}$$

where κ_{EX} and κ_{IM} are steady-state ratios of exports and imports to GDP. The eigenvalue $(1/\beta - \psi) < 1$ ensures stationarity.

3.9 Exchange rate pass-through to sectoral costs

Defining **import centrality**

$$\mathcal{M}_i \;\equiv\; \big[(\mathbf{I} - \mathbf{\Omega}^H)^{-1}\mathbf{\Omega}^F \mathbf{1}\big]_i, \tag{23}$$

the total exposure of sector i's inflation to an FX/import-price shock under a peg is proportional to $\mathcal{M}_i$. Sectors with higher import centrality transmit more FX shocks to domestic prices.

3.10 Policy rules

Three alternative regimes are implemented by replacing one block in Dynare:

Regime A — Free float (benchmark):

$$\widehat{\imath}_t = \phi_\pi \pi_t^{DC} + \phi_y \widetilde{y}_t. \tag{24}$$

$\widehat{s}_t$ is endogenous (determined by UIP).

Regime B — Hard peg:

$$\widehat{s}_t = 0 \quad \forall\, t. \tag{25}$$

$\widehat{\imath}_t$ is residually determined by UIP (**??**).

Regime C — Managed float:

$$\widehat{\imath}_t = \phi_\pi \pi_t^{DC} + \phi_y \widetilde{y}_t + \phi_s \widehat{s}_t. \tag{26}$$

$\phi_s > 0$ implies partial exchange-rate stabilisation; $\phi_s \to \infty$ approaches a peg.

3.11 Exogenous shock processes

$$\widehat{a}_{it} = \rho_a\, \widehat{a}_{i,t-1} + \varepsilon_{it}^a, \qquad i = 1, 2, 3, \tag{27}$$

$$\widehat{p}_t^{F*} = \rho_{pF}\, \widehat{p}_{t-1}^{F*} + \varepsilon_t^{pF}, \tag{28}$$

$$\widehat{D}_t^* = \rho_D\, \widehat{D}_{t-1}^* + \varepsilon_t^D. \tag{29}$$

3.12 Variable summary

Variable	Type	Description
$\pi_{1t}, \pi_{2t}, \pi_{3t}$	Jump	Sectoral inflation (Resource, Manuf., Services)
$\widetilde{y}_t$	Jump	Output gap
$\widehat{s}_t$	Jump (float) / Exog.=0 (peg)	Nominal exchange rate (dep.>0)
$\widehat{\imath}_t$	Jump (float) / Residual (peg)	Domestic interest rate
p_{1t}, p_{2t}, p_{3t}	State (endogenous)	Sectoral price levels
$\widehat{b}_t^*$	State (endogenous)	Net foreign assets
$\widehat{w}_t$	Static	Nominal wage
$\pi_t^C, \widehat{p}_t^C, \pi_t^{DC}$	Static	CPI inflation, CPI level, DC index
$\widehat{a}_{1t}, \widehat{a}_{2t}, \widehat{a}_{3t}$	State (exogenous)	Sectoral TFP
$\widehat{p}_t^{F*}$	State (exogenous)	Import price shock
$\widehat{D}_t^*$	State (exogenous)	Foreign demand shock

4 Key Analytical Results

Lemma 1 (Output gap and markups)**.** *$(\gamma + \varphi)\widetilde{y}_t = -\sum_i \lambda_{D,i}\mu_{it}$, where $\mu_{it} = p_{it} - mc_{it}$ is sector i's log markup.*

Lemma 2 (Markup–inflation link)**.** $\mu_{it} = -\dfrac{1-\widehat{\delta}_i}{\widehat{\delta}_i}(\pi_{it} - \beta \mathbb{E}_t \pi_{i,t+1})$.

Proposition 1 (Open-economy divine coincidence)**.** *The DC index* (**??**) *satisfies*

$$\pi_t^{DC} = \beta\,\mathbb{E}_t \pi_{t+1}^{DC} + \kappa^{DC} \widetilde{y}_t, \quad \kappa^{DC} = \frac{\gamma + \varphi}{\sum_k \lambda_{D,k}(1 - \widehat{\delta}_k)/\widehat{\delta}_k}, \tag{30}$$

*with **no residual:** neither domestic TFP shocks $\widehat{\mathbf{a}}_t$ nor the exchange rate $\widehat{s}_t$ nor import prices $\widehat{p}_t^{F*}$ appear.*

Proof sketch: *Embed each imported input as a fully-flexible upstream node ($\widehat{\delta}_M = 1$) in an augmented network. Fully flexible nodes receive zero weight in the DC index ($w_M = 0$) because $(1 - \widehat{\delta}_M)/\widehat{\delta}_M = 0$. Rubbo's Proposition 1 applies verbatim to the augmented network.* □

Corollary (FX regime comparison). Under the float (**??**) with $\phi_\pi > 1$, the exchange rate absorbs import-price shocks: $\widehat{s}_t \approx -\widehat{p}_t^{F*}$, so all sectoral inflations converge to zero and $\widetilde{y}_t \to 0$. Under the peg (**??**), import-price shocks pass through into domestic inflation with magnitude proportional to the import centrality $\mathcal{M}_i$ of each sector.

5 Calibration

The adjusted Calvo parameters (**??**) are $(\widehat{\delta}_1, \widehat{\delta}_2, \widehat{\delta}_3) = (0.693,\ 0.336,\ 0.079)$.

6 Dynare Implementation

`open_economy_network.mod` implements the *original nonlinear* structural model – Cobb-Douglas cost minimization, the exact recursive Calvo pricing problem, the CRRA/Frisch household FOCs, CES consumption aggregation, and UIP with a debt-elastic premium – not the log-linearized system of Section 3. Dynare computes the non-stochastic steady state numerically (`initval/steady;`) and then perturbs around it (`stoch_simul(order=2,pruning,...)`); the log-linear system above is kept only as an independent theory benchmark to sanity-check Dynare's own order-1 decision rules against. Two modelling choices worth flagging:

- **Output gap.** Rather than solving a parallel flex-price shadow economy, $\widetilde{y}_t$ uses Lemma 1 in closed form: $\widetilde{y}_t = \sum_i \lambda_{D,i}\big[\log(Y_{it}/\bar{Y}_i) - \log A_{it}\big]$, an exact implication of the model rather than an approximation.
- **$\widehat{\delta}_i$ is not a primitive here.** The raw Calvo reset probabilities $\delta_i = (0.75, 0.50, 0.25)$ enter the nonlinear recursion directly; $\widehat{\delta}_i$ (eq. **??**) is computed only to build the theory-implied DC-index weights fed into the Taylor rule, since it is itself a log-linearization artifact of the exact reset-price recursion (proofs.tex, Calvo section).

Regimes are switched via the `@#define REGIME` macro at the top of the file (`"float"`, `"peg"`, `"managed"`). `code/run_all_regimes.m` runs all three automatically (by generating one temporary `.mod` file per regime and calling Dynare on each) and exports IRFs, the derived network objects (λ_D, import centrality, $\widehat{\delta}$, DC weights) and order-2 variances to `results/*.csv`. `code/analysis.py` then loads those CSVs to build the IRF, network/centrality, DC-insulation and welfare-decomposition figures referenced in `OpenEconomy_Networks_FX.pptx`; see that file's docstring for the full pipeline. Superseded illustrative code that solved a hand-rolled approximate linear system (not a genuine rational-expectations solution) previously lived in `code/model.py`.

For reference, the file skeleton (see the repository for the full, current version – it changes independently of this write-up):

Listing 1: open_economy_network.mod — structure only, see repo for current content

```
// var Y1 Y2 Y3 L1 L2 L3 P1 P2 P3 PI1 PI2 PI3 MC1 MC2 MC3 PSTAR1-3
```

```
//      X1_i X2_i (Calvo recursion) C CH CF PH PC PIC Ltot W
//      S PF PFH BSTAR I EX IM DSTAR A1 A2 A3 piDC y_gap GDP;
// varexo eps_a1 eps_a2 eps_a3 eps_pF eps_D;
//
// model;
//   MC_i           = Cobb-Douglas nominal marginal cost
//   X1_i, X2_i     = exact Calvo recursive aggregators (real MC/P, not MC)
//   PSTAR_i        = optimal reset price = (eps/(eps-1))*X1_i/X2_i
//   1 = (1-delta_i)*PI_i^(eps-1) + delta_i*PSTAR_i^(1-eps)   [Calvo law]
//   L_i            = Cobb-Douglas labour demand
//   Y_i            = domestic C + intermediate demand + exports (mkt clearing)
//   PH, PC, CH, CF = Cobb-Douglas / CES consumption aggregation
//   Euler, labour supply (nonlinear CRRA/Frisch FOCs)
//   UIP with debt-elastic premium; NFA current-account identity
//   EX, IM         = export demand / total import demand
//   policy rule    = float | peg | managed (regime-dependent)
//   piDC, y_gap    = DC index (Prop. 1 weights); output gap (Lemma 1, closed form)
//   AR(1) shock processes in logs
// end;
```

(Retained below for historical reference: the very first hand-linearized draft of this file, before it was rewritten as the nonlinear structural model described above.)

Listing 2: Superseded: original hand-linearized draft

```
// =========================================================
// Open Economy Production Network NK Model
// Rubbo (2024) extended to Small Open Economy
// 3 sectors: Resource (1), Manufacturing (2), Services (3)
//
// REGIMES (uncomment one):
// @#define REGIME = "float"    // Free float (default)
// @#define REGIME = "peg"      // Hard peg
// @#define REGIME = "managed"  // Managed float
@#define REGIME = "float"
// =========================================================

// ---------------------------------------------------------
// VARIABLES
// ---------------------------------------------------------
var
    // Sectoral inflation
    pi1 pi2 pi3

    // Aggregate inflation measures
    piDC piCPI

    // Output gap and interest rate
    y_gap i_hat

    // Exchange rate (log deviation, depreciation = positive)
    s_hat

    // Sectoral price levels (log deviation)
    p1 p2 p3
```

```

    // Consumer price level (log deviation)
    pC_hat

    // Real wage (log deviation)
    w_hat

    // Net foreign assets (log deviation)
    bstar

    // TFP shocks (sectoral)
    a1 a2 a3

    // Import price shock
    pF

    // Foreign demand shock
    D_star
    ;

// ---------------------------------------------------------
// EXOGENOUS SHOCKS
// ---------------------------------------------------------
varexo eps_a1 eps_a2 eps_a3 eps_pF eps_D;

// ---------------------------------------------------------
// PARAMETERS
// ---------------------------------------------------------
parameters
    // Preferences
    BETA GAMMA VARPHI

    // Adjusted Calvo parameters (Rubbo eq. after (14))
    DHAT1 DHAT2 DHAT3

    // Labour shares
    ALPHA1 ALPHA2 ALPHA3

    // Domestic input-output shares
    OH21      // Manuf. buys from Resource (domestic)
    OH32      // Services buys from Manuf. (domestic)

    // Import shares (total per sector)
    OF1 OF2 OF3

    // Final consumption shares (domestic sectors)
    BH1 BH2 BH3

    // Total import share in final consumption
    BF_TOT

    // DC index weights
    WDC1 WDC2 WDC3

    // Open economy
    PSI       // NFA premium
    THETA_S   // export price elasticity
    KAP_EX    // exports/GDP
```

```
    KAP_IM    // imports/GDP
    NU_D      // foreign demand sensitivity

    // Policy
    PHI_PI PHI_Y PHI_S

    // Shock persistence
    RHO_A RHO_PF RHO_D
    ;

// ----------------------------------------------------------
// PARAMETER VALUES
// ----------------------------------------------------------

// Preferences
BETA    = 0.99;
GAMMA   = 1.00;
VARPHI  = 2.00;

// Adjusted Calvo (delta_hat from eq. (2))
// delta = [0.75, 0.50, 0.25], beta = 0.99
// dhat_i = delta_i*(1-beta*(1-delta_i))/(1-beta*delta_i*(1-delta_i))
DHAT1   = 0.693;
DHAT2   = 0.336;
DHAT3   = 0.079;

// Labour shares (alpha_i = 1 - sum_j(omega_H_ij + omega_F_ij))
ALPHA1  = 0.70;   // Resource:  1 - 0.00 - 0.30
ALPHA2  = 0.70;   // Manuf.:    1 - 0.20 - 0.10
ALPHA3  = 0.70;   // Services:  1 - 0.25 - 0.05

// Domestic I-O shares
OH21    = 0.20;   // Manuf. <- Resource (domestic)
OH32    = 0.25;   // Services <- Manuf. (domestic)

// Total import shares per sector
OF1     = 0.30;   // Resource
OF2     = 0.10;   // Manufacturing
OF3     = 0.05;   // Services

// Final consumption shares
BH1     = 0.05;
BH2     = 0.15;
BH3     = 0.80;
BF_TOT  = 0.10;

// DC index weights  w_i propto lambda_D_i*(1-dhat_i)/dhat_i
// lambda_D = [0.12, 0.35, 0.80] (domestic supplier centrality)
// raw weights:
//   w1 = 0.12*(1-0.693)/0.693 = 0.0532
//   w2 = 0.35*(1-0.336)/0.336 = 0.6916
//   w3 = 0.80*(1-0.079)/0.079 = 9.328
//   total = 10.073
WDC1    = 0.0053;
WDC2    = 0.0686;
WDC3    = 0.9261;

// Open economy
```

```
PSI      = 0.020;
THETA_S  = 2.00;
KAP_EX   = 0.85;
KAP_IM   = 0.45;
NU_D     = 0.30;

// Policy
PHI_PI   = 1.50;
PHI_Y    = 0.50;
PHI_S    = 0.30;   // managed float only

// Shock persistence
RHO_A    = 0.90;
RHO_PF   = 0.85;
RHO_D    = 0.80;

// -------------------------------------------------------------
// MODEL BLOCK
// -------------------------------------------------------------
model(linear);

//--------------------------------------------------------------
// [1] Sectoral Phillips curves  (eqs. 3-5 in paper)
// pi_it = beta*(1-dhat_i)*E[pi_{i,t+1}]
//       + dhat_i*[alpha_i*w + omega_H_ij*p_j - p_i(-1)
//                 + omega_F_i*(s+pF) - a_i]
// Note: p_jt = p_j(-1) + pi_j  is substituted inline
//--------------------------------------------------------------

// Resource (no domestic inputs)
pi1 = BETA*(1-DHAT1)*pi1(+1)
    + DHAT1*(   ALPHA1*w_hat
              + OF1*(s_hat + pF)
              - a1
              - p1(-1)
            );

// Manufacturing (buys from Resource domestic)
pi2 = BETA*(1-DHAT2)*pi2(+1)
    + DHAT2*(   ALPHA2*w_hat
              + OH21*(p1(-1) + pi1)
              + OF2*(s_hat + pF)
              - a2
              - p2(-1)
            );

// Services (buys from Manufacturing domestic)
pi3 = BETA*(1-DHAT3)*pi3(+1)
    + DHAT3*(   ALPHA3*w_hat
              + OH32*(p2(-1) + pi2)
              + OF3*(s_hat + pF)
              - a3
              - p3(-1)
            );

//--------------------------------------------------------------
// [2] Price level laws of motion  (eq. 6)
//--------------------------------------------------------------
```

```
p1 = p1(-1) + pi1;
p2 = p2(-1) + pi2;
p3 = p3(-1) + pi3;

//------------------------------------------------------------
// [3] Consumer price level  (eq. 9)
//------------------------------------------------------------
pC_hat = BH1*p1 + BH2*p2 + BH3*p3 + BF_TOT*(s_hat + pF);

//------------------------------------------------------------
// [4] CPI inflation  (eq. 8)
//------------------------------------------------------------
piCPI = BH1*pi1 + BH2*pi2 + BH3*pi3
      + BF_TOT*((s_hat + pF) - (s_hat(-1) + pF(-1)));

//------------------------------------------------------------
// [5] Real wage  (eq. 7, labour supply)
// w_hat = pC_hat + (gamma+varphi)*y_gap
//------------------------------------------------------------
w_hat = pC_hat + (GAMMA + VARPHI)*y_gap;

//------------------------------------------------------------
// [6] DC inflation index  (eq. 10, Proposition 1)
//------------------------------------------------------------
piDC = WDC1*pi1 + WDC2*pi2 + WDC3*pi3;

//------------------------------------------------------------
// [7] IS curve  (eq. 11)
// y_gap = E[y_gap(+1)] - (1/gamma)*(i_hat - E[piCPI(+1)])
//       + nu_D * D_star
// (natural rate r_nat = 0 in deviations from SS)
//------------------------------------------------------------
y_gap = y_gap(+1) - (1/GAMMA)*(i_hat - piCPI(+1)) + NU_D*D_star;

//------------------------------------------------------------
// [8] UIP  (eq. 12)
// i_hat = -psi*bstar + E[s(+1)] - s
//------------------------------------------------------------
i_hat = -PSI*bstar + s_hat(+1) - s_hat;

//------------------------------------------------------------
// [9] NFA accumulation  (eq. 13)
// bstar = (1/beta - psi)*bstar(-1)
//       + kap_EX*(-theta_s*s + D_star)
//       - kap_IM*(pF + y_gap)
// Eigenvalue = 1/beta - psi = 0.990 < 1 (stationary)
//------------------------------------------------------------
bstar = (1/BETA - PSI)*bstar(-1)
      + KAP_EX*(-THETA_S*s_hat + D_star)
      - KAP_IM*(pF + y_gap);

//------------------------------------------------------------
// [10] Policy rule (regime-dependent)
//------------------------------------------------------------
@#if REGIME == "float"
    // Free float: Taylor rule on DC index
    i_hat = PHI_PI*piDC + PHI_Y*y_gap;

```

```
@#elseif REGIME == "peg"
    // Hard peg: exchange rate fixed
    s_hat = 0;
    // i_hat determined residually by UIP (already in eq. 8)

@#elseif REGIME == "managed"
    // Managed float: Taylor rule + FX stabilisation
    i_hat = PHI_PI*piDC + PHI_Y*y_gap + PHI_S*s_hat;

@#endif

//------------------------------------------------------------
// [11] Shock processes  (eqs. 15-17)
//------------------------------------------------------------
a1 = RHO_A*a1(-1)  + eps_a1;
a2 = RHO_A*a2(-1)  + eps_a2;
a3 = RHO_A*a3(-1)  + eps_a3;
pF = RHO_PF*pF(-1) + eps_pF;
D_star = RHO_D*D_star(-1) + eps_D;

end;

// ----------------------------------------------------------
// STEADY STATE
// (All log-deviations are zero by construction)
// ----------------------------------------------------------
initval;
    pi1 = 0; pi2 = 0; pi3 = 0;
    piDC = 0; piCPI = 0;
    y_gap = 0; i_hat = 0; s_hat = 0;
    p1 = 0; p2 = 0; p3 = 0;
    pC_hat = 0; w_hat = 0; bstar = 0;
    a1 = 0; a2 = 0; a3 = 0;
    pF = 0; D_star = 0;
end;

// ----------------------------------------------------------
// SHOCKS (unit standard deviations)
// ----------------------------------------------------------
shocks;
    var eps_a1  = 0.01^2;   // 1% TFP shock, sector 1 (Resource)
    var eps_a2  = 0.01^2;   // 1% TFP shock, sector 2 (Manuf.)
    var eps_a3  = 0.01^2;   // 1% TFP shock, sector 3 (Services)
    var eps_pF  = 0.01^2;   // 1% import price shock
    var eps_D   = 0.01^2;   // 1% foreign demand shock
end;

// ----------------------------------------------------------
// SOLUTION AND OUTPUT
// ----------------------------------------------------------
stoch_simul(
    order        = 1,      // linear model
    periods      = 0,      // no simulation; compute IRFs only
    irf          = 40,     // 40-quarter IRF horizon
    graph_format = pdf,
    nograph      = 0,
    noprint      = 0
);
```

```
// Print key model moments
disp('===␣DC␣index␣weights␣===');
disp([WDC1 WDC2 WDC3]);

disp('===␣Import␣centrality␣(network␣exposure␣to␣FX␣shock)␣===');
// M_i = [((I-OmegaH)^{-1}) * OF] -- computed externally from network
    matrices
// Values: [0.30, 0.16, 0.09] for Resource, Manuf., Services
```

(Historical) switching regimes, peg residual, welfare snippet. The three paragraphs and Matlab snippet below describe the superseded linear draft (variable names `s_hat`, `i_hat`, `pi1-3`). In the current nonlinear file the same ideas hold with the new variable names (`S`, `I`, `PI1-3`): switch regimes via the `@#define REGIME` macro (or let `code/run_all_regimes.m` do it for all three); under the peg, `I` is residual, pinned down by the UIP equation; and the welfare computation is now `compute_welfare()` in `code/analysis.py`, reading `results/variances.csv` and `results/network_objects.csv` exported by `run_all_regimes.m` – same formula (eq. **??** below), but evaluated on the true nonlinear model's order-2 unconditional variances rather than a linearized approximation.

$$\mathcal{W} = \tfrac{1}{2}\left[(\gamma+\varphi)\,\mathrm{Var}(\widetilde{y}_t) + \sum_i \lambda_{D,i}\varepsilon_i \frac{1-\widehat{\delta}_i}{\widehat{\delta}_i}\,\mathrm{Var}(\pi_{it})\right]. \tag{31}$$

References

- Galí, J. and Monacelli, T. (2005). Monetary policy and exchange rate volatility in a small open economy. *RES* 72(3), 707–734.
- Pasten, E., Schoenle, R. and Weber, M. (2017). Price rigidity and the origins of aggregate fluctuations. NBER WP 23750.
- Qiu, Z., Wang, Y., Xu, L. and Zanetti, F. (2025/26). Monetary policy in open economies with production networks. *JME*, forthcoming. CEPR DP19828.
- Rubbo, E. (2024). Networks, Phillips curves, and monetary policy. *Econometrica* 91(4), 1417–1455.
- Schmitt-Grohé, S. and Uribe, M. (2003). Closing small open economy models. *JIE* 61(1), 163–185.
- Smets, F. and Wouters, R. (2003). An estimated dynamic stochastic general equilibrium model of the euro area. *JMCB* 35(5), 1123–1162.

Table 1: Calibration — 3-sector oil-exporter (quarterly)

Parameter	**Value**	**Description**	**Source**
Preferences			
β	0.99	Discount factor	Standard
γ	1.00	Risk aversion / inverse EIS	Standard
φ	2.00	Inverse Frisch elasticity	**?**
η	1.50	Home–foreign substitution	**?**
ω	0.70	Home bias in consumption	Data average
Domestic input-output matrix $\mathbf{\Omega}^H$			
ω_{21}^H	0.20	Manuf. buys 20% from Resource	BEA I-O
ω_{32}^H	0.25	Services buys 25% from Manuf.	BEA I-O
Imported input matrix $\mathbf{\Omega}^F$			
ω_1^F	0.30	Resource: 30% imported machinery	SOE data
ω_2^F	0.10	Manuf.: 10% imported components	SOE data
ω_3^F	0.05	Services: 5% indirect imports	SOE data
Labour shares $\alpha_i = 1 - \sum_j(\omega_{ij}^H + \omega_{ij}^F)$			
α_1	0.70	Resource	Derived
α_2	0.70	Manufacturing	Derived
α_3	0.70	Services	Derived
Final consumption shares			
$\beta_1^H, \beta_2^H, \beta_3^H$	0.05, 0.15, 0.80	Domestic sectors	Expenditure surve
$\beta^{F,\text{tot}}$	0.10	Total import share in consumption	Trade data
Pricing (Calvo)			
δ_1	0.75	Resource: flexible (commodity prices)	**?**
δ_2	0.50	Manufacturing: intermediate	**?**
δ_3	0.25	Services: sticky	**?**
ε_i	8.00	Within-sector substitution elasticity	Standard NK
Open economy			
ψ	0.020	NFA debt-elastic premium (ensures stationarity)	**?**
θ^*	2.00	Export price elasticity	**?**
κ_{EX}	0.85	Exports/GDP ratio	Oil exporter data
κ_{IM}	0.45	Imports/GDP ratio	Oil exporter data
ν_D	0.30	Foreign demand sensitivity	Calibrated
Monetary policy			
ϕ_π	1.50	Taylor rule: inflation	**?**
ϕ_y	0.50	Taylor rule: output gap	**?**
ϕ_s	0.30	Managed float: FX weight	Calibrated
Shock persistence			
ρ_a	0.90	TFP	Standard
ρ_{pF}	0.85	Import price	Commodity price
ρ_D	0.80	Foreign demand	Standard
Derived network objects			
$\lambda_{D,1}, \lambda_{D,2}, \lambda_{D,3}$	0.12, 0.35, 0.80	Domestic supplier centrality	Eq. (**??**)
$w_1^{DC}, w_2^{DC}, w_3^{DC}$	0.005, 0.069, 0.926	DC index weights	Eq. (**??**)
κ^{DC}	0.298	DC Phillips curve slope	Eq. (**??**)
$\mathcal{M}_1, \mathcal{M}_2, \mathcal{M}_3$	0.30, 0.16, 0.09	Import centrality	Eq. (**??**)